
Types of Motors

There are 3 types of motors commonly used in Arduino-based projects:

DC Motors

Simple motors that turn continuously when an electric current is passed through them. 
They are simple to control because the speed with which they turn can be controlled by the 
amount of voltage across their terminals.

Stepper Motors

Stepper motors also turn when a voltage is applied to them, but they only turn by a very 
small step, e.g., 1.8 degrees, every time the voltage is applied. Stepper motors require much 
more complex electronics but are commonly used for precise movement.

Servo Motors

Servo motors typically combine DC motors (or sometimes stepper motors) with position 
sensors and control electronics so that they can be instructed to move to a particular 
position and hold it.

What is a Servo Motor?

A servo motor is a general class of device that combines a motor with sensors to 
maintain position or rotation speed.

A hobby (or RC) servo is a common type of servo motor combines a DC motor, a gear 
train (to reduce the speed of the motor), a potentiometer (to sense the position of the 
output shaft), an integrated circuit (to monitor and adjust the motor based on the 
output shaft), and an output shaft bearing.

The three wires that stick out from the motor casing include: one is for power, one is 
for ground, and one is a control input line.

The shaft of the servo can be positioned to specific angular positions by sending a coded 
signal. As long as the coded signal exists on the input line, the servo will maintain the 
angular position of the shaft. If the coded signal changes, then the angular position of the 
shaft changes.

Some hobby servos are modified to maintain an angular velocity, instead of position, in 
response to the input signal.

Circuit #8: Single Servo

#include <Servo.h>
Servo servo1;

void setup() { servo1.attach(9); }

void loop() {
  int position;
  servo1.write(90);  delay(1000); // Tell servo to go to 90 degrees
  servo1.write(180); delay(1000); // Tell servo to go to 180 degrees
  servo1.write(0);   delay(1000); // Tell servo to go to 0 degrees

  for (position = 0; position < 180; position += 2) {
    servo1.write(position); delay(20); // Move to next position
  }

  for (position = 180; position >= 0; position -= 1) {                                
    servo1.write(position); delay(20); // Move to next position
  }
}

Circuit #12: Spinning a Motor



Circuit #12: Circuit Diagram

RedBoard

Resistor (330 ohm)
(Orange-Orange-Brown)

Transistor
(P2N2222AG)

DC Motor
Diode
(1N4148)

GND
(Ground)

5 volt

PIN 9



RedBoard

Resistor (330 ohm)
(Orange-Orange-Brown)

Transistor
(P2N2222AG)

DC Motor
Diode
(1N4148)

GND

5 volt

PIN 9

Circuit #12: Why do we need a transistor?

The Arduino pins are strong enough to light small LEDs (up to 40 milliAmps), but 
they're not strong enough to run motors and other power-hungry parts.

The motor in your kit needs 50-100mA.

A transistor is a solid-state switch. When we give it a small amount of current, it can 
switch a much larger current.

The transistors in your kit (2N2222) can switch up to 200mA.



Circuit #12: Why do we need a diode?

When the motor is spinning and suddenly turned off, the magnetic field inside it 
collapses, generating a voltage spike. This can damage the transistor.

To prevent this, we use a "flyback diode", which diverts the voltage spike "around" the 
transistor.

Remember that current only flows through a diode in one direction, in circuit diagrams this 
is in the direction the arrow is pointing.

This means that when there is a voltage spike the current will be allowed to flow back away 
from the transistor, but when the motor is running the current won’t flow through the diode 
and will only flow through the motor.

Circuit #12

const int motorPin = 9;
void setup() {
  pinMode(motorPin, OUTPUT);
  Serial.begin(9600);
}

void loop() {
  motorOnThenOff();
  // motorOnThenOffWithSpeed();
  // motorAcceleration();
  // serialSpeed();
}



Circuit #12 (cont.)

void motorOnThenOff() {
  int onTime = 3000;  // milliseconds to turn the motor on
  int offTime = 3000; // milliseconds to turn the motor off
  
  digitalWrite(motorPin, HIGH); // turn the motor on (full speed)
  delay(onTime);                // delay for onTime milliseconds
  digitalWrite(motorPin, LOW);  // turn the motor off
  delay(offTime);               // delay for offTime milliseconds
}



Circuit #12 (cont.)

void motorOnThenOffWithSpeed() {
  int Speed1 = 200;  // between 0 (stopped) and 255 (full speed)
  int Time1 = 3000;  // milliseconds for speed 1
  
  int Speed2 = 50;   // between 0 (stopped) and 255 (full speed)
  int Time2 = 3000;  // milliseconds to turn the motor off
  
  analogWrite(motorPin, Speed1);  // turns the motor On
  delay(Time1);                   // delay for onTime milliseconds
  analogWrite(motorPin, Speed2);  // turns the motor Off
  delay(Time2);                   // delay for offTime milliseconds
}



Circuit #12 (cont.)

void motorAcceleration() {
  int speed;
  int delayTime = 20; // milliseconds between each speed step
  
  // accelerate the motor
  for (speed = 0; speed <= 255; speed++) {
    analogWrite(motorPin,speed); // set the new speed
    delay(delayTime);            // delay between speed steps
  }
  
  // decelerate the motor
  for (speed = 255; speed >= 0; speed--) {
    analogWrite(motorPin,speed); // set the new speed
    delay(delayTime);            // delay between speed steps
  }
}



Circuit #12 (cont.)

void serialSpeed() {
  int speed;
  Serial.println("Type a speed (0-255) into the box above,");
  Serial.println("then click [send] or press [return]");
  Serial.println();

  while (true) {
    while (Serial.available() > 0) {
      speed = Serial.parseInt();
      speed = constrain(speed, 0, 255);
      Serial.print("Setting speed to ");
      Serial.println(speed);
      analogWrite(motorPin, speed);
    }
  }
}



Relays

A relay is a electrically-controlled mechanical switch. It can control much more voltage 
and current than an Arduino pin, or the smaller types of transistor, like the one included 
in the Inventor’s Kit.

Because the relay needs more power to switch than an Arduino pin can provide, we'll use a 
transistor in Circuit #13 to drive the relay in exactly the same way we used a transistor to 
drive a motor in Circuit #12.

A relay is an excellent way to control a lightbulb, coffee maker, or other high-power 
device. Despite being mechanical, rather than solid state, relays are very reliable and 
can be switched millions of times before failure.

The disadvantage of using a mechanical relay is that it is slower to switch than a solid 
state switch, like a transistor. Alternatives, such as devices called MOSFETs, are forms of 
transistors that can handle higher voltage and current in DC circuits.



Inside a Relay

Types of Relays



Circuit #13: Relays

Circuit #13: Relays

const int relayPin = 2;     // use this pin to drive the transistor
const int timeDelay = 1000; // delay in ms for on and off phases

void setup() {
  pinMode(relayPin, OUTPUT);  // set pin as an output
}

void loop() {
  digitalWrite(relayPin, HIGH);  // turn the relay on
  delay(timeDelay);              // wait for one second
  digitalWrite(relayPin, LOW);   // turn the relay off
  delay(timeDelay);              // wait for one second
} 


